Mutiple DC Solution Determination using VHDL-AMS

Joachim Haase
Fraunhofer-Institute for Integrated Circuits/
Branch Lab Design Automation EAS Dresden

Gerd Pönisch
Dresden University of Technology
Institute of Numerical Mathematics

Manfred Uhle
Weierstrass Institute for
Applied Analysis and Stochastics Berlin

Problem and Proposed Approach

The investigation of nonlinear resistive networks is one of the fundamental tasks in circuit simulation. Relative difficult problems that have to be solved are:

- Finding the multiple DC solutions of circuits as flip-flops, Schmitt-triggers, and negative resistance circuits.
- Calculation of multivalued transfer characteristics of nonlinear resistive networks (e.g. of Schmitt-triggers).
- Calculation of the turning points of multivalued transfer characteristics (e.g. in order to determine hysteresis or pull-in voltages in electromechanical systems).

Continuation techniques based on homotopy methods are typically applied to solve these problems whereas these methods are implemented in customized simulators.

Proposed way in the presented paper:

- Direct application of the homotopy idea to the network analysis problem by creating a modified and augmented network that is similar to the original one.
- Constitutive relation of the modified subnets can be expressed using a behavioral description language, e. g. VHDL-AMS.
- Solution of the modified network using an available VHDL-AMS simulation engine. From the solution of the modified and augmented network we get the solution we are interested in.

An advantage of the suggested approach is that no special simulation engine is necessary. All features of available VHDL-AMS simulation engines can be used.

Homotopy Methods

- For the analysis of nonlinear resistive networks a system of nonlinear equations has to be solved: $F(x) = 0$,where $F : R^n \rightarrow R^n$
- Continuation methods based on a homotopy function $H : R^n \times R \rightarrow R^n$ with the following characteristics:
  - The solution $x_0 \in R^n$ of $H(x_0, \lambda_0) = 0$ can be determined for a given $\lambda_0 \in R$.
  - For a special $\lambda_f \in R$ the solution $x_f \in R^n$ of $H(x_f, \lambda_f) = 0$ is also a solution of F, i. e. $F(x_f) = 0$.
- The solution is calculated by following the solution curve $L = \{ (x, \lambda) \in R^n \times R \mid H(x, \lambda) = 0 \}$ from (x_0, λ_0) to (x_f, λ_f).
- L can be traced for instance by solving the following DAE

$$H(x, \lambda) = 0$$
$$\sum_{i \in I} x_i'(t)^2 + \lambda'(t)^2 = 1, \quad \text{where } (x(0), \lambda(0)) = (x_0, \lambda_0).$$

I is a set of essential variables for curve tracing. t is a pseudo arc length of the solution curve L.

Network Formulation

- Assume a resistive network N is given. We construct a modified network N_λ that is close-by the original. λ is an additional quantity. N is in accordance with F. N_λ corresponds to the homotopy function H.
- To calculate the DC operating point, the modification can be done for instance by
  - Replacement of independent voltage and current sources with values V_{cl} and I_{cl} resp. by sources with values $\lambda \cdot V_{cl}$ and $\lambda \cdot I_{cl}$ (*source stepping*).
  - Replacement of open branches by additional current sources with values $(1-\lambda) \cdot I_{c0}$ (*artificial input current source stepping*).
  - ... *G-min stepping*, ...

  Operating points can be detected in these cases for λ=1.
- N_λ is augmented by a subsystem that realizes the equation $\sum_{i \in I} pr_i\, x'(t)^2 + \lambda'(t)^2 = 0$. The output λ of this subsystem drives the modified sources and conductances.
- The augmented network can be evaluated with the time domain simulation algorithm of the simulation engine. The condition λ=1 can be detected using the 'ABOVE attribute of VHDL-AMS.
- Determination of transfer characteristics and turning points can be done in an equivalent manner.

Network Modifications (N_λ)

```
entity css is
 generic (vc0    : real := 0.0);
 port(terminal p, m      : electrical;
      quantity lambda    : in real);
end entity css;

architecture a0 of css is
 signal   i0  : real;
 quantity v     across i through p to m;
begin

 process(domain) is begin
   i0 <= i;
 end process;

 if domain = quiescent_domain use
    v  == vc0;
 else
    i  == (1.0-lambda)*i0;
 end use;

end architecture a0;
```



Artificial input current source

```
entity trace is
 generic  (N    : natural := 1;
           ta   : real    := 1.0e-3;
           tb   : real    := 5.0e-3);
 port (quantity x       : in  real_vector (1 to N);
       quantity lambda : out real);
end entity trace;

architecture a1 of trace is
 quantity lambda_dot    : real;
 quantity x_help, x_dot : real_vector (1 to N);

 function factor (t: real; ta : real; tb : real)
   return real is
   variable th     : real;
 begin
   th := (t-ta)/(tb-ta);
   return (3.0*th*th - 2.0*th*th*th);
 end function factor;

begin
 lambda_dot == lambda'dot;
 x_help     == x;
 x_dot      == x_help'dot;
 if now < ta use
    lambda  == now;
 elsif now < tb use
    1.0     == lambda_dot*lambda_dot
              + factor(now,ta,tb)*(x_dot*x_dot);
 else
    1.0 == lambda_dot*lambda_dot + x_dot*x_dot;
 end use;
end architecture a1;
```



Model realizes $\sum_{i \in I} x_i'(t)^2 + \lambda'(t)^2 = 1$

Multiple DC Solutions (Two-Tunnel-Diode Circuit)

```
entity detect is
 port (quantity lambda    : in real;
       terminal p_voltage : electrical;
       signal   s_voltage : out real := 0.0);
end entity detect;

architecture one of detect is
 quantity v across p_voltage;
begin
 process (lambda'above(1.0)) is
 begin
   s_voltage <= v;
 end process;
end architecture one;
```



Modified and augmented network

Subcircuit to detect λ = 1 (driven by λ, observes voltage at p_voltage)

Calculation of Multivalued Transfer Characteristics (Schmitt-Trigger)




```
entity vss is
 port (terminal p, m    : electrical;
       quantity lambda : in real);
end entity vss;

architecture a0 of vss is
 quantity v across i through p to m;
begin
 v  == lambda;
end architecture a0;
```



Input voltage source for calculation of multivalued transfer characteristics



Result of transient simulation and transfer chatacteristic $v_{out}(v_{in})$

Determination of Turning Points (Pull-in Voltage Calculation)

```
entity detect is
 port (terminal p_voltage : electrical;
       signal   s_voltage : out real := 0.0;
       signal   stop      : out boolean);
end entity detect;

architecture turning of detect is
 quantity v        across p_voltage;
begin
 process is
 begin
  stop <= false;
  wait on domain;
  while true loop
    wait on v'dot'above(0.0);
    s_voltage <= v;
    stop      <= true;
  end loop;
 end process;
end architecture turning;
```





Transfer characteristic:
Displacement as function of v;
Turning point
delivers the pull-in voltage

Subcircuit to detect the turning point

Fraunhofer Institut Integrierte Schaltungen
IIS

TECHNISCHE UNIVERSITÄT DRESDEN

WIAS
Weierstraß-Institut für Angewandte Analysis und Stochastik